

Automatisch telefoneren via UNIFACE

Naar een idee van Leendert Koning

Bewerkt door Klaas Robers

Het interface-systeem UNIFACE kan ingezet worden voor het automatisch bellen van telefoonnummers zodat u, nadat u het telefoonnummer in een database-programma heeft gevonden, met één druk op de knop de bij het nummer behorende persoon door de computer kunt laten bellen. Mits zijn telefoonnummer is ingevoerd natuurlijk. Leendert Koning stuurde ons het ontwerp voor deze UNIFACE-toepassing die hij met veel plezier gebruikt op de P2000. Met wat aanpassingen kan echter ook de MSX of de PC gebruikt worden.


Het principe

Het bellen verloopt via twee relais die voor de kiespulsen zorgen. Bij het bellen wordt eerst relais 1 bekrachtigd, er gaat een stroompje lopen en er wordt een lijn van de centrale toegewezen. Vervolgens worden via relais 2 draad a en b van de telefoon kortgesloten, wat aangeeft dat er een nummer gekozen gaat worden (dit gebeurt normaal door de kiesschijf naar rechts te draaien), gevolgd door een stel snel op elkaar volgende pulsen. Die pulsen bestaan uit 'even geen weerstand' (de relais trekken aan) en 'even een oneindige weerstand' (de relais allebei open). Dit gebeurt net zo vaak als het te draaien cijfer. Dan volgt nog even een kortsluitpuls en dan weer alleen de weerstand die door relais 1 bekrachtigd wordt. Daarna volgt het volgende cijfer. Als het hele nummer gedraaid is moet de hoorn opgepakt worden en dan moet ook relais 1 weer geopend worden.

N.B. Uiteraard moet het signaal van de uitgangen wel eerst versterkt worden!

Dit kan gebeuren met een eenvoudige transistor. De schakeling van de relais wordt dan als weergegeven in figuur 1.

Afhankelijk van de spanning die nodig is voor de relais kan +5 Volt via UNIFACE uit de computer worden gehaald of er moet uit een aparte voeding +12 Volt of misschien wel +24 Volt worden betrokken. Als diode is vrijwel elke diode bruikbaar in dit geval.



Figuur 1: schakeling van de relais.

Voorbeeld van een kiesprogramma op de P2000

Het voorbeeldprogramma voor de P2000 kiest een telefoonnummer dat op het toetsenbord wordt ingegeven, maar kan bijvoorbeeld ook aan een aangepaste versie van het Supersnelzoekboek gekoppeld worden.

Aanpassing voor PC en MSX

Het voorbeeldprogramma doet dit alles voor u. Het is opgebouwd in subroutines zodat het gemakkelijker te begrijpen is en ook kan worden ingebouwd in een adressen programma. De ON ERROR GOTO zit er in om te voorkomen dat bij fouten de telefoonlijn lange tijd blijft opgenomen.


Het programma is gemaakt voor de P2000, maar het kan gemakkelijk worden aangepast voor MSX of PC. In regel 10 staat het poortadres van UNIFACE. Voor de P2000 is dit &H60, maak hiervan voor de MSX: &H30 en voor de PC: SH310.

In regel 20 staat een T die de snelheid van het programma bepaalt. Voor de P2000 is 40 de juiste waarde. Voor een PC (niet op turbo) blijkt daar 50 te moeten staan. De juiste waarde is eenvoudig experimenteel te vinden. Regel 30 roept een wacht-routine van 10 seconden aan.

Verander T net zo lang tot het wachten precies 10 seconden duurt. Daarna mag regel 30, en 500 en 510 natuurlijk, worden verwijderd.

PTT

De relaischakeling, die het opnemen en kiezen van de telefoon simuleert, werkt netjes precies als een echte telefoon. Met de privatisering van de PTT is men en stuk soepeler geworden met wat er wel en wat er niet mag worden aangesloten. Het risico met deze schakeling dat er iets fout gaat met de telefoon is niet zo groot. Door de relais is




Telefoonstekker

de computer gescheiden van de telefoonlijn EN DAT MOET OOK! De "aarde" van UNIFACE mag nergens worden doorverbonden met de schakelcontacten die naar de a en b van de telefoon gaan.

Er zijn (bijvoorbeeld bij V&D) telefoonstekers te koop waar je achterin weer een telefoonsteker kunt prikken. Deze zijn bij uitstek geschikt voor deze toepassing. De stekker die naar de relais kontakten gaat wordt dan gewoon onder de telefoon stekker gestoken. Zo is in no-time alles weer origineel.

Figuur 2



```

10 U=&H60:REM UNIFACE poortadres
20 T=40:REM Wachtijd voor pulsen
30 GOSUB 500:REM Test wachtijd
50 OUT U,0:ON ERROR GOTO 90
60 GOSUB 100:REM Dit is het programma
90 OUT U,0:ON ERROR GOTO 0:END
95 :
100 REM *** Telefoonnummer kiezen ***
110 INPUT"Telefoonnummer";T$
120 OUT U,2:REM Relais 1 neemt telefoon op
130 FOR I=1 TO 20*T:NEXT:REM Wacht op kiestoon
140 GOSUB 200:REM Kies procedure
150 PRINT:PRINT"Klaar !"
160 PRINT"Pak de hoorn en geef 'ENTER'"
170 INPUT X$:REM wacht op ENTER
180 RETURN
190 :
200 REM *** Draai het nummer in T$ ***
210 FOR C=1 TO LEN(T$)
220 A$=MID$(T$,C,1)
230 IF A$="-" THEN GOSUB 300:GOTO 280
240 A=VAL(A$):IF A$="0" THEN A=10
250 IF A=0 THEN GOTO 280
260 PRINT A$;
270 GOSUB 400:REM Draai het cijfer (A)
280 NEXT:RETURN:REM Klaar, nummer gedraaid
290 :
300 REM *** Wacht op interlokaal toon ***
310 PRINT A$;:FOR I=1 TO 40*T:NEXT:RETURN
400 REM *** Draai een cijfer ***
410 OUT U,6:REM Relais 1 en 2 aangetrokken
420 FOR I=1 TO 10*T:NEXT:REM Kiesschijf opdraaien
430 FOR L=1 TO A:REM Maak A pulsen
440 OUT U,0:FOR I=1 TO T:NEXT:REM Puls
450 OUT U,6:FOR I=1 TO T:NEXT:REM Pauze
460 NEXT:OUT U,2:REM Kiesschijf afgelopen
470 FOR I=1 TO 10*T:NEXT:REM Wacht na cijfer
480 RETURN
490 :
500 PRINT "*** Wacht 10 seconden ***"
510 FOR I=1 TO 200*T:NEXT:RETURN

```